

Limity funkcí

Limity funkcí se počítají podobně jako limity posloupností. Jenom je více možností vzhledem k tomu, že limitní body jsou i vlastní a je možné se k těmto bodům blížit zprava nebo zleva.

Podívejme se nejdříve na bezproblémové limity.

```
In[1]:= Limit[(1 - Cos[x]) / x^2, x -> 0]
```

```
Out[1]= 1/2
```

```
In[2]:= Limit[(1 + a / x)^x, x -> Infinity]
```

```
Out[2]= e^a
```

```
In[3]:= Limit[x * Sin[1 / x], x -> 0]
```

```
Out[3]= 0
```

```
In[4]:= Limit[((1 + x)^a - 1) / x, x -> 0]
```

```
Out[4]= a
```

```
In[5]:= Limit[Tan[2 * x] * Tan[Pi / 4 - x], x -> Pi / 4]
```

```
Out[5]= 1/2
```

```
In[6]:= Limit[Sin[Sqrt[x + 1]] - Sin[Sqrt[x]], x -> Infinity]
```

```
Out[6]= 0
```

```
In[7]:= Limit[x * Floor[x], x -> 0]
```

```
Out[7]= 0
```

```
In[8]:= Limit[x^x, x -> 0]
```

```
Out[8]= 1
```

```
In[9]:= Limit[x^(1 / (Log[x])), x -> 0]
```

```
Out[9]= e
```

```
In[10]:= Limit[(Sqrt[1 + x] - (1 + x)^(1 / 3)) / x, x -> Infinity]
```

```
Out[10]= 0
```

```
In[11]:= Limit[(1 + 2 / x)^Sin[x], x -> 0]
```

```
Out[11]= 1
```

Pokud limita neexistuje a v okolí limitního bodu osciluje, vypíše *Mathematica* limitu oscilačního intervalu.

```
In[12]:= Limit[Sin[1 / x], x -> 0]
```

```
Out[12]= Interval[{-1, 1}]
```

```
In[13]:= Limit[(x + 2) * Sin[1 / x], x -> 0]
```

```
Out[13]= Interval[{-2, 2}]
```

Pokud obsahuje funkce parametr a limity jsou různé v závislosti na tomto parametru, limita se nespočte, je nutné přidat podmínky na parametr.

```
In[14]:= Limit[(a^x - 1) / x, x -> 0]
```

```
Out[14]= Log[a]
```

```
In[15]:= Limit[a^x, x -> Infinity]
```

```
Out[15]= Limit[a^x, x -> ∞]
```

```
In[16]:= Limit[a^x, x -> Infinity, Assumptions -> a > 1]
```

```
Out[16]= ∞
```

```
In[17]:= Limit[a^x, x -> Infinity, Assumptions -> 0 < a < 1]
```

```
Out[17]= 0
```

```
In[18]:= Limit[a^x, x -> Infinity, Assumptions -> a == 1]
```

```
Out[18]= 1
```

Bohužel, *Mathematica* má jednu nectnost. Limitu totiž počítá (kromě limity v nekonečnu) vždy zprava. Pokud s tím někdo nepočítá, nůž dospět ke špatným výsledkům, protože může brát výsledek (který je uveden bez varování) jako oboustrannou limitu. Teprve přidáním podmínky se dozvíme správnou situaci.

```
In[19]:= Limit[Sign[x], x -> 0]
```

```
Out[19]= 1
```

```
In[20]:= Limit[Sign[x], x -> 0, Direction -> 1]
```

```
Out[20]= -1
```

```
In[21]:= Limit[Sign[x], x -> 0, Assumptions -> x < 0]
```

```
Out[21]= -1
```

Ani přidáním podmínek nemusíme dostat správný výsledek.

```
In[22]:= Limit[Sqrt[x], x -> 0, Direction -> 1]
```

```
Out[22]= 0
```

```
In[23]:= Limit[Sqrt[x], x -> 0, Assumptions -> {x < 0, Element[x, Reals]}]
```

```
Out[23]= 0
```


```
In[24]:= Limit[Sqrt[-x], x -> 0, Assumptions -> Element[x, Reals]]
```

```
Out[24]= 0
```

U limit funkcí definovaných po částech *Mathematica* nespočte limity funkcí definovaných pomocí podmínek `/;`. U funkcí definovaných pomocí `Piecewise` nebo `Which` je vše v pořádku.

```
In[25]:= Clear[f]; f[x_] := Piecewise[{{x^2, x < 0}, {x + 2, 0 < x < 3}}, 1]
```

In[26]:= `Plot[f[x], {x, -2, 4}]`



In[27]:= `Limit[f[x], x -> 0]`


Out[27]= 2

In[28]:= `Limit[f[x], x -> 0, Direction -> 1]`

Out[28]= 0

In[29]:= `Clear[g]; g[x_] := x^2 /; x < 0; g[x_] := x + 2 /; 0 < x < 3; g[x_] := 1 /; x > 3`

In[30]:= `Plot[g[x], {x, -2, 4}]`




In[31]:= `Limit[g[x], x -> 1]`

Out[31]= `Limit[g[x], x -> 1]`

In[32]:= `Clear[h]; h[x_] := Which[x < 0, x^2, 0 < x < 3, x + 2, x > 2, 1]`

```
In[33]:= Plot[h[x], {x, -2, 4}]
```



```
In[34]:= Limit[h[x], x → 0]
```

Out[34]= 2

```
In[35]:= Limit[h[x], x → 0, Direction → 1]
```

Out[35]= 0